TwoPDETestb.xmcd. September 11, 2009.. This worksheet modifies the PDE Solve Quicksheet is solve the linearized partial differential equations describing undrained, adiabatic shear of a veloci strengthening fault zone. The equations are written in terms of the gradients of pore pressure and temperature.

Adds results for FD solution to previous worksheets.

In these equations,  $\Phi$  is the gradient of pore pressure and  $\Psi$  is gradient of temperature in non-dimensional form and perturbations from the uniform solution. cth and chy are thermal and hydraulic diffusivities, z is a parameter that gives the magnitude of the velocity strengthening (she rate strengthening) and Hc is a measure of magnitude of the shear heating (all in nondimensiona form). Representative values are given below.

cth := 
$$10^{-2}$$
 Hc := 0.3 chy :=  $1 \cdot 10^{-4}$  z := 40

Linearized solution for temperature and pressure (and their gradients) is unstable (grows exponentially) for wavenumbers less than a critical value given by (Critical value of the velocity is slightly different because of the effect of the time dependence of the homogeneous solution.

Ncrit := 
$$\frac{1}{\pi} \cdot \sqrt{\frac{z \cdot Hc}{(cth + chy)}}$$
 Ncrit = 10.972  
N1 := floor(Ncrit) floor(Ncrit) = 10  
N2 := ceil(Ncrit) ceil(Ncrit) = 11  
 $\Phi 0 := 1$   $\Psi 0 := 0$   
P0(y) :=  $\Phi 0 \cdot \sin(\pi \cdot \text{Nnum} \cdot \text{y})$  T0(y) :=  $\Psi 0 \cdot \sin(\pi \cdot \text{Nnum} \cdot \text{y})$   
Given  
 $\Psi_t(y,t) = cth \cdot \Psi_{yy}(y,t) + Hc \cdot z \cdot \Phi(y,t)$ 

 $\Phi_{t}(y,t) = \left( chy \cdot \Phi_{yy}(y,t) + cth \cdot \Psi_{yy}(y,t) \right) + Hc \cdot z \cdot \Phi(y,t)$ 

Initial conditions (for single Fourier mode):

$$\Phi(y,0) = PO(y)$$

 $\Psi(y,0) = T0(y)$ 

Boundary conditions (since solution is anti-symmetric about y = 0, bc's given for half the domain

$\Psi(L,t) = 0$	$\Phi(L,t)=0$
$\Psi(0,t) = 0$	$\Phi(0,t)=0$

Solve over the ranges 0 to L in y and 0 to T in t.

$$\begin{pmatrix} \Phi \text{num} \\ \Psi \text{num} \end{pmatrix} \coloneqq \text{Pdesolve} \begin{bmatrix} \Phi \\ \Psi \end{bmatrix}, \text{y}, \begin{pmatrix} 0 \\ L \end{pmatrix}, t, \begin{pmatrix} 0 \\ T \end{bmatrix}, \text{spacepts, timepts} \end{bmatrix}$$

## Exact (i.e., analytical) Solution

Exact solution (verified and tested in LinearSolFulla.xmcd. Solutions are of the form exp(pt)  $sin(N^*\pi Y)$  (because of the boundary conditions). Then p must satisfy a quadratic equation with the following coefficients::

$$Bc(n) := \frac{1}{2} \cdot \left[ (\pi \cdot n)^{2} (cth + chy) - z \cdot Hc \right] \qquad Cc(n) := (\pi \cdot n)^{4} \cdot (cth \cdot chy)$$
  
Discrim(n) :=  $\sqrt{(Bc(n))^{2} - Cc(n)}$ 

Two solutions are:

pplus := -Bc(Nnum) + Discrim(Nnum) pminus := -Bc(Nnum) - Discrim(Nnum)

pplus = -1.177 + 0.796i pminus = -1.177 - 0.796i

$$\Phi \text{plus} \coloneqq \frac{\text{pplus} + \text{cth} \cdot (\text{Nnum} \cdot \pi)^2}{\text{pplus} - \text{pminus}} \cdot \left[ \Phi 0 - \frac{\Psi 0}{\text{Hc} \cdot z} \cdot \left[ \text{pminus} + \text{cth} \cdot (\text{Nnum} \cdot \pi)^2 \right] \right]$$

 $\Phi$ minus :=  $\Phi 0 - \Phi$ plus

$$\Psi \text{plus} := \frac{\text{Hc} \cdot z \cdot \Phi \text{plus}}{\left[\text{pplus} + \text{cth} \cdot (\text{Nnum} \cdot \pi)^2\right]} \qquad \Psi \text{minus} := \frac{\text{Hc} \cdot z \cdot \Phi \text{minus}}{\left[\text{pminus} + \text{cth} \cdot (\text{Nnum} \cdot \pi)^2\right]}$$

 $\Psi$ time(s) :=  $\Psi$ plus·exp(pplus·s) +  $\Psi$ minus·exp(pminus·s)

 $\Phi$ time(s) :=  $\Phi$ plus·exp(pplus·s) +  $\Phi$ minus·exp(pminus·s)

 $\Psi$ anal $(y, s) := \Psi$ time $(s) \cdot sin(Nnum \cdot \pi \cdot y)$   $\Phi$ anal $(y, s) := \Phi$ time $(s) \cdot sin(Nnum \cdot \pi \cdot y)$ 

## **Explicit Finite Difference Scheme**

nypts := 100 ky := 0.. nypts yL := 1.0 deltay :=  $\frac{yL}{nypts}$ 

$$dtmin1 := 0.5 \cdot \frac{deltay^2}{chy} \qquad \qquad dtmin2 := 0.5 \cdot \frac{deltay^2}{cth}$$

Time step must be smaller than dtmin1 and dtmin2 for convergence.

## **Compare Analytical and Numerical Solutions**


$L \equiv 1$	$T \equiv 3$	spacepts $\equiv 100$	timepts $\equiv 100$	Nnum $\equiv 12$
y := 0,0	.01L	t := 0,0.01T		

deltat1 := 0.00125

ntpts := 2400 kt := 0.. ntpts Tfinal := deltat1.ntpts SOL := Sol2(ntpts, deltat1) Psol :=  $SOL_0$  Tsol :=  $SOL_1$ 

Compare time history at ypeak := 0.04. This is near the first peak in the sine wave, i.e., at  $\frac{1}{2 \cdot \text{Nnum}} = 0.042$ 

 $\frac{20}{15}$   $\frac{\Psi num(ypeak, t)}{\Psi anal(ypeak, t)}$   $\frac{10}{10}$   $\frac{1}{5}$   $\frac$ 



Note agreement of analytical and finite difference solutions, but divergence of the PDESolve solution as the time approaches 3. Increasing the number of time points in PDESolve does not seem to mitigate the problem (I have tried the same number of points as in the FD scheme).